# Worlds within Worlds: An Introduction to Cellular Automata

*By Jeremiah P. Reilly and Nicholas B. Tufillaro*

The Christmas-tree pattern presented in Fig. 1 would make a nice holiday-season greeting card. This particular pattern is an example of a cellular automaton, a model universe that physicists and computer scientists are exploring in order to understand our world better. Essentially, a cellular automaton is a lattice of identical sites where the value at each site is determined by some rule. Although the rule is usually simple, the resulting pattern may be quite complex, as our Christmas tree illustrates. In fact, the rules are so simple that inexpensive home computers can quickly generate the Christmas-tree pattern or millions of other pictures. The universe of cellular automata is huge, and you can easily get lost in its beauty and complexity.

Before we learn how to build a cellular automaton on our home computer, let's look at a few more pictures. Looking at the pyramid in Fig. 1, we first notice that there is regularity to the pattern. The elements grow outward. Next, perhaps, we notice the triangles within triangles within triangles. Further, there is a certain symmetry and order. It turns out that the development of this cellular automaton is deterministic and predictable. Such predictability is not usual, however.

The multiple symmetries may not be apparent at first glance. To illustrate the inherent symmetry, imagine cutting out the triangle in Fig. 1. Next, photograph this section and enlarge it fourfold so that the image is the same size as the original pattern. We would discover that the enlargement resembles the entire original pattern. This simple expansion is an example of "self-similarity." A figure is self-similar if we can find scaling laws—enlargements, for example—with which we can recover the information in the entire figure from just a portion. Of course, in the cellular automaton we have presented, we must ignore the discrete nature of the elements (the pixel size of the computer screen). The size of the lattice in a computer-generated cellular automaton is finite. We are limited in how small a region we can pick for our enlargement.



**Fig. 1. Mod 2 cellular automaton with single initial site. The accompanying BASIC program, with small modification, will run on most home computers.**



**Jeremiah P. Reilly** *is a free-lance writer with an interest in science writing, primarily physics and genetics, for technical and lay readers (2225 Spruce Street, Philadelphia, PA 19103). He first studied chaos in Hesiod's* Theogony. *He is also a guitar maker intent on understanding the production of sound in the guitar. Jeremiah's undergraduate training was in classical Greek at Swarthmore College. In addition, he holds a M.A. in ancient history from the University of Pennsylvania and a Master of Public Policy from Harvard University's John F. Kennedy School of Government.*

**Table I. Program for Fig. 1.**

```
10   REM Cellular Automata Program
20   REM Rule: Mod 2. Initial Value: single site
30   REM Copyright 1985 by Nicholas B. Tufillaro
40   INTEGER i, j, k
50   REM Site arrays. R is current row, Q is previous
     row.
60   INTEGER R(512), Q(512)
70   XMAX = 512
80   YMAX = 256
90   GRAPHICS
100  SCALE 0, XMAX − 1, YMAX − 1, 0
110  Q(255) = 1
120  MOVE 255, 0
130  PLOT 255, 0
140  FOR n = 1 TO YMAX − 1
150       FOR j = 0 TO XMAX − 1
160            i = j − 1
170            IF i = −1 THEN i = XMAX − 1
180            k = j + 1
190            IF k = XMAX THEN k = 0
200            REM Cellular Automata Rule
210            site = (Q(i) + Q(k)) MOD 2
220            R(j) = site
230            MOVE j, n
240            IF site = 1 THEN PLOT j, n
250       NEXT j
260       MAT Q = R
270  NEXT n
280  END
```

Although the pattern in Fig. 1 is two-dimensional, it is composed of generations of one-dimensional cellular automata, each succeeding one placed below its predecessor. Each line in the cellular automaton represents one generation. A cellular automaton rule determines the transition from one generation to the next. It is, if you like, the physical law that governs the system. Thus, a cellular automaton is a dynamic system — a system in motion.

The rule for the cellular automaton in Fig. 1 is exceedingly simple. If the value of the sites adjacent to a given site are both black or both white, then in the next generation the site will be white. If only one adjacent site is black, then the given site will be black in the next generation. This is called the "monogamy" or "mod 2" rule. Shown in Table I is a simple BASIC program to generate the mod 2 cellular automaton. All of the pictures seen here were generated on an Atari 800XL computer or on a Macintosh.

To illustrate the infinite variation possible in cellular automata, let's make a minimal modification to the cellular automaton in Fig. 1. We will retain the same initial condition: a single occupied cell in the first generation (top line). However, we will modify the cellular automaton rule (see the accompanying program in Table II). The resulting cellular automaton is depicted in Fig. 2. Notice the surprising differences. Most obvious is the lack of regularity. The development of this cellular automaton is chaotic. Whereas the behavior of the cellular automaton governed by the monogamy rule was predictable, the state of any future generation in Fig. 2 is unknowable without explicitly calculating all of the preceding generations. We have just illustrated a fascinating feature of cellular automata: although deterministic — remember that the cellular automaton rule governs the dynamics of the system at each generation — the behavior is unpredictable.

We can also generate apparently random behavior from a cellular automaton with the mod 2 rule. Let's alter the initial conditions in the first generation while retaining the mod 2 rule. Instead of starting with a single initial site, we generate a random selection of initial sites. The resulting cellular automaton is shown in Fig. 3; the complex behavior

**Nick Tufillaro** *received his B.A. in physics from Reed College in Portland, OR, and is currently completing his Ph.D. in physics at Bryn Mawr College (Bryn Mawr, PA 19010). His interests include nonlinear dynamics, bicycling, and children. During 1988–89, he was a Fulbright scholar at the department of physics, University of Otago, Dunedin, New Zealand.*





**Fig. 2. Chaotic cellular automaton (Rule 30) with single initial site and accompanying BASIC program.**

is striking. We can also provide random initial conditions to the cellular automaton governed by the "chaotic rule" (Fig. 2). Figure 4 shows the resulting cellular automaton.

**Table II. Program for Fig. 2.**

```
10   REM Cellular Automata Program
20   REM Rule: 30. Initial Value: single site
30   REM Copyright 1985 by Nicholas B. Tufillaro
40   INTEGER i, j, k
50   REM Site arrays. R is current row, Q is previous
     row.
60   INTEGER R(512), Q(512)
70   XMAX = 512
80   YMAX = 256
90   GRAPHICS
100  SCALE 0, XMAX – 1, YMAX – 1, 0
110  Q(255) = 1
120  MOVE 255, 0
130  PLOT 255, 0
140  FOR n = 1 TO YMAX – 1
150       FOR j = 0 TO XMAX – 1
160            i = j – 1
170            IF i = – 1 THEN i = XMAX – 1
180            k = j + 1
190            IF k = XMAX THEN k = 0
200            REM Cellular Automata Rule 30
210            site = (Q(i) + MAX (Q(j), Q(k)))
               MOD 2
220            R(j) = site
230            MOVE j, n
240       IF site = 1 THEN PLOT j, n
250       NEXT j
260       MAT Q = R
270  NEXT n
280  END
```

A given cellular automaton is determined by two ingredients. The first is the initial conditions, that is, the initial value at each site in the lattice. The second ingredient is the cellular automaton rule. The rule defines the future value at a given site and other sites based on the previous value at that site and other sites, usually neighboring sites. This method of sampling values is called a nearest neighbor interaction.

Cellular automata are classified according to the dimensionality of the lattice. If all sites lie on a line, a cellular automaton is one dimensional, or linear. We have looked at one-dimensional cellular automata in our examples. Each line in these examples represents a new generation. To be precise, the cellular automaton at each generation lies on a line. Displaying each new generation under the previous generation is called a space-time graph. The space axis is horizontal, while the time axis is vertical. These cellular automata may look two dimensional, but they are not.

Cellular automata are planar, or two dimensional, if all the cells lie in a plane. So far, scientists have studied only one-dimensional automata in detail. Recently, researchers have begun to explore cellular automata of two and more dimensions. The hope is that two-dimensional cellular automata will provide good models for certain crystal formations, such as the growth of snowflakes. Multidimensional cellular automata can be quite complex. Efficient examination of such complex cellular automata of higher dimensions is beyond the capabilities of most current microcomputers, unless they get help from special-purpose hardware.




**Fig. 3. Mod 2 cellular automaton with random initial conditions.**

Aside from the beauty and intriguing intricacy of cellular automata, why are scientists investigating their behavior? It is necessary to reiterate a basic aspect of cellular automata: the rules determine, in fact they comprise, the dynamics of the system. Even a simple rule can produce unpredictable behavior. Unexpectedly, this deterministic randomness is paralleled by physical systems, such as turbulent fluid flow and dendritic crystal growth, which are also deterministic and unpredictable. Stephen Wolfram (formerly a physicist at the Institute for Advanced Study at Princeton, NJ, and now director of the Center for Complex Systems Research, University of Illinois at Urbana-Champaign), has recently suggested that many systems in nature are like cellular automata in that the only way to




**Fig. 4. Random initial conditions supplied to the cellular automaton in Fig. 2. Rule 30 still governs.**

examine the behavior of the system is to simulate its dynamics – specifically on a computer.

A general-purpose cellular automata simulation system for the Atari 800XL or the Macintosh is available on floppy disk from the authors. Please send a 3.5-in floppy and a self-addressed, stamped envelope with your request. All the pictures shown here were created with the Atari Cellular Automata Simulation System or Automata on the Macintosh. ♦

# What is Physics?

*Here are some definitions given by students to David W. Nevins of Alfred State College, Alfred, NY 14802.*

- The ability to fall down and blame it on gravity or the study of similar material.
- The in-depth study of physical laws. It's also required.
- The laws that allow us to do what we do!
- After taking it once before, my first impression would be let's get through it in one piece. Physics is a scary subject inflicted upon engineering students to confuse and disorient them.
- Physics is the study of work and energy; it is also going to kill my GPA.
- My impression of physics is that it is a difficult course in which 50 percent of the students fail.
- The study of things that have been constant for years and are all written in reference books. Physics is the cause of many an unhappy student in high school and college. If a train is traveling 75 mph in a southeasterly direction, how many of its passengers would have a working knowledge of physics and how many would want one?
- I never took physics before, but I have heard some rumors about it. I hope they are not true. ♦